

# SMARKY NEWS

No 15

Le 15 janvier 1982

## FICHIERS BASIC

### 1. Définition

On appelle fichier un ensemble d'informations mémorisées sur un média (écran, ruban de papier, cartes perforées, imprimante, cassette, disquette, disque ou toute autre mémoire de masse).

Dans ce texte, nous développerons uniquement la gestion de fichiers mémorisés sur disquettes et de fichiers édités par l'imprimante.

### 2. Opérations de travail avec des fichiers

Le travail sur un fichier se déroule suivant le schéma suivant:

- A. Création ou ouverture
- B. Positionnement à l'intérieur du fichier
- C. Accès (lecture ou écriture)
- D. Test éventuel de fin de fichier
- E. Fermeture.

Les opérations A et E n'ont lieu qu'une fois pour un fichier déterminé, alors que les opérations B, C et D peuvent se répéter plusieurs fois.

### 3. Fichiers disques

Il existe deux types de fichiers-disques :

- 1.- Les fichiers séquentiels
- 2.- Les fichiers à accès aléatoires

Ces deux genres de fichiers se traitent par des opérations ou lignes de commande assez ressemblantes, mais les deux types ne peuvent absolument pas être confondus.

Le langage BASIC offre la possibilité de travailler sur plusieurs fichiers simultanément. Il faut donc préciser pour chaque fichier un numéro de canal (de 0 à 7).

#### 3.1 Fichier séquentiel

Ce type de fichier est caractérisé par le fait que les différentes informations se présentent de manière successive.

On commence par accéder à la première, puis à la seconde, et ainsi de suite jusqu'à la dernière.

Il est à remarquer que contrairement au fichier aléatoire, chaque enregistrement peut avoir une longueur variable et il n'est pas nécessaire de connaître à priori le nombre d'enregistrements que l'on désire faire (écriture).

#### 3.1.1 Création ou ouverture

Ligne de commande : OPEN FILE(c,m),A\$

avec c = No. de canal

m = mode d'ouverture

1 pour création en écriture

3 pour ouverture en lecture

A\$= nom (selon règles des chaînes de caractères)

#### 3.1.2 Positionnement

La commande de positionnement dans un fichier séquentiel est automatique, le pointeur passant à la position suivante après chaque accès.

#### 3.1.3 Accès au fichier

A.- En écriture:

Ligne de commande: PRINT FILE(c),C\$

pour une écriture d'information dans le fichier

avec c = No. de canal

C\$= chaîne de caractères écrits.

B.- En lecture:

Ligne de commande : INPUT FILE(c),B\$

pour une lecture d'information contenue dans le fichier

avec c = No. de canal,

B\$= chaîne de caractères lus.

Remarque: l'instruction INPUT FILE(c),LINE B\$ permet de conserver les espaces.

1. - Exemple de fichier séquentiel (ESC = fin de l'écriture).

```
100 PRINT "CREATION DU FICHIER"
110 INPUT "Nom du fichier ";N$
120 PRINT "OUVERTURE DU FICHIER EN ECRITURE"
130 OPEN FILE (1,1),N$
140 PRINT "ECRITURE"
150 LET I=I+1: PRINT "Fiche No ";I;
160 INPUT LINE A$
170 ON ESC THEN 200
180 PRINT FILE (1),A$
190 GOTO 150
200 PRINT "FERMETURE DU FICHIER"
210 CLOSE FILE (1)
220 PRINT
230 PRINT "OUVERTURE DU FICHIER EN LECTURE"
240 OPEN FILE (1,3),N$
250 PRINT "LECTURE"
260 INPUT FILE (1), LINE A$
270 PRINT A$
280 IF EOF(1) THEN 300
290 GOTO 260
300 PRINT "FERMETURE DU FICHIER"
310 CLOSE FILE (1)
320 ON ESC THEN STOP
330 PRINT "DESTRUCTION DU FICHIER"
340 DELETE N$
350 END
```

2. - Exemple de fichier aléatoire (ESC = fin de la lecture).

```
100 PRINT "CREATION DU FICHIER"
110 INPUT "NOM DU FICHIER ";N$
120 INPUT "Nombre de fiches ";N
130 INPUT "Nombre de caractères d'une fiche ";L
140 PRINT "OUVERTURE EN ECRITURE"
150 OPEN FILE (1,0),N$,L,N
160 PRINT "ECRITURE"
170 FOR I=0 TO N-1
180 SPOS (1,I)
190 PRINT "Fiche No ";I+1;
200 INPUT LINE A$
210 LET X=LEN(A$)
220 IF X>L THEN PRINT "La fiche contient ";ABS(L-X);" caractères en trop": GOTO 190
230 PRINT FILE (1),A$
240 NEXT I
250 PRINT "FERMETURE DU FICHIER"
260 CLOSE FILE (1)
270 PRINT
280 PRINT "OUVERTURE DU FICHIER EN LECTURE"
290 OPEN FILE (1,2),N$
300 PRINT "LECTURE"
310 PRINT N;"Fiches enregistrées - Pointeur sur la fiche No (ESC = FIN)";: INPUT I
320 IF I>N OR I<=0 THEN PRINT "MORS DU FICHIER": BEEP : GOTO 310
330 ON ESC THEN 380
340 SPOS (1,I-1)
350 INPUT FILE (1), LINE A$
360 PRINT A$
370 GOTO 310
380 PRINT "FERMETURE DU FICHIER"
390 CLOSE FILE (1)
400 ON ESC THEN STOP
410 PRINT "DESTRUCTION DU FICHIER"
420 DELETE N$
430 END
```

### 3. - Exemple de fichier "imprimante".

```
100 PRINT "EXEMPLE DE FICHIER IMPRIMANTE"  
110 PRINT "Impression des nombres entiers de 1 à 10"  
120 PRINT : PRINT "OUVERTURE DU FICHIER $LP EN ECRITURE"  
130 OPEN FILE (1,1), "$LP"  
140 PRINT : PRINT "TRANSFERT DES INFORMATIONS SUR L'IMPRIMANTE"  
150 FOR A=1 TO 10  
160 LET A$=NUM$(A)  
170 PRINT FILE (1),A$  
180 NEXT A  
190 PRINT : PRINT "FERMETURE DU FICHIER $LP"  
200 CLOSE FILE (1)  
210 END
```

### 4. - Exemple de manipulation de fichiers.

```
100 PAGE  
110 PRINT "CREATION D'UN FICHIER AVEC DES INFORMATIONS"  
120 PRINT "CONTENUES DANS DEUX AUTRES FICHIERS"  
130 PRINT "(Les fichiers F1 et F2 contiendront 3 informations)"  
140 PRINT : PRINT "CREATION ET OUVERTURE DU FICHIER F1 EN ECRITURE"  
150 OPEN FILE (1,1), "F1"  
160 PRINT "ECRITURE DU FICHIER F1"  
170 FOR I=1 TO 3  
180 PRINT "Enregistrement No. ";I;" du fichier F1": INPUT A$  
190 PRINT FILE (1),A$  
200 NEXT I  
210 PRINT : PRINT "CREATION ET OUVERTURE DU FICHIER F2 EN ECRITURE "  
220 OPEN FILE (2,1), "F2"  
230 PRINT "ECRITURE DU FICHIER F2"  
240 FOR I=1 TO 3  
250 PRINT "Enregistrement No. ";I;" du fichier F2": INPUT B$  
260 PRINT FILE (2),B$  
270 NEXT I  
280 PRINT "FERMETURE DES FICHIERS F1 ET F2"  
290 CLOSE FILE (1)  
300 CLOSE FILE (2)  
310 PAGE : PRINT "CREATION ET OUVERTURE DU FICHIER F3 EN ECRITURE"  
320 OPEN FILE (3,1), "F3"  
330 PRINT "ECRITURE DU FICHIER F3"  
340 OPEN FILE (1,3), "F1"  
350 OPEN FILE (2,3), "F2"  
360 FOR I=1 TO 3  
370 INPUT FILE (1),A$  
380 INPUT FILE (2),B$  
390 LET C$=A$+SPACE$(1)+B$  
400 PRINT FILE (3),C$  
410 NEXT I  
420 CLOSE  
430 PRINT "OUVERTURE DU FICHIER F3 EN LECTURE"  
440 OPEN FILE (3,3), "F3"  
450 PRINT "LECTURE DU FICHIER F3"  
460 INPUT FILE (3), LINE C$  
470 IF EOF(3) THEN 500  
480 PRINT C$  
490 GOTO 460  
500 PRINT "FERMETURE DU FICHIER F3"  
510 CLOSE FILE (3)  
520 PRINT : PRINT "DESTRUCTION DES FICHIERS F1, F2 ET F3"  
530 DELETE "F1"  
540 DELETE "F2"  
550 DELETE "F3"  
560 END
```

### 3.1.4 Test de fin de fichier (EOF)

Utilisable en cas de lecture seulement, cette instruction permet de tester si le fichier est terminé.

Ligne de commande: IF EOF(c) THEN ...

### 3.1.5 Fermeture du fichier

Un travail de lecture ou d'écriture doit se terminer par une instruction de fermeture du fichier.

Ligne de commande : CLOSE ferme tous les canaux ou  
CLOSE FILE(c) avec c = No. de canal.

### 3.2 Fichier à accès aléatoire

Ce type de fichier est caractérisé par le fait qu'on peut accéder aux différentes informations de manière libre au moyen d'une instruction de pointage en un endroit quelconque du fichier. D'autre part, chaque enregistrement aura la même longueur puisque, à priori, lors de la création, il faut déclarer le nombre et la longueur de chaque enregistrement.

#### 3.2.1 Création ou ouverture

La création a lieu lorsque le fichier n'existe pas encore.

L'ouverture a lieu lorsque le fichier existe déjà.

Ligne de commande : OPEN FILE(c,m),A\$(l,n)

avec c = No. du canal

m = mode d'ouverture

= 0 pour création (l et n sont obligatoires)

= 2 pour ouverture (l et n sont omis)

A\$= nom (selon règles des chaînes de caractères)

l = longueur des enregistrements

n = nombre d'enregistrements.

#### 3.2.2 Positionnement

Il s'agit de choisir l'emplacement où se fera la lecture ou l'écriture.

Ligne de commande : SPOS(c,r)

avec c = No. de canal

r = rang

#### 3.2.3 Accès

A.- En écriture:

Ligne de commande : PRINT FILE(c),C\$

pour une écriture d'information dans le fichier

avec c = No. de canal

C\$= chaîne de caractères écrits.

B.- En lecture:

Ligne de commande : INPUT FILE(c),B\$

pour une lecture d'information contenue dans le fichier  
avec c = No. de canal

B\$= chaîne de caractères lus.

Remarque: l'instruction INPUT FILE(c),LINE B\$ permet de conserver les espaces.

#### 3.2.4 Test de fin de fichier (EOF)

Utilisable seulement en cas de lecture séquentielle de tout le fichier, cette instruction permet de tester si le fichier est terminé.

Ligne de commande : IF EOF(c) THEN ...

#### 3.2.5 Fermeture du fichier

Un travail de lecture ou d'écriture doit se terminer par une instruction de fermeture du fichier.

Ligne de commande : CLOSE ou CLOSE FILE(c)

avec c = No. de canal.

### 4. Fichier imprimante

Ce fichier est en fait un cas particulier de fichier séquentiel dans lequel il n'est possible que d'écrire.

L'information est transmise directement de l'ordinateur au papier.

L'opération de lecture n'existe donc pas.

L'imprimante est un fichier séquentiel dont le nom est "\$LP".

#### 4.2.1 Ouverture

voir fichier séquentiel

#### 4.2.2 Positionnement

voir fichier séquentiel

#### 4.2.3 Accès

voir fichier séquentiel

#### 4.2.4 Test de fin de fichier

inexistant car on ne lit pas d'information de l'imprimante.

#### 4.2.5 Fermeture

voir fichier séquentiel

